
ABSTRACT
Communication is the basic foundation of all relationships, including our SAS® relationship with the server, PC,
or mainframe. To communicate more efficiently ~ and to increasingly automate your SAS world ~ you will
want to learn how to transform static code into dynamic code that automatically re-creates the static code, and
then executes the re-created static code automatically. Our presentation highlights the powerful partnership
that occurs when dynamic code is creatively combined with a dynamic FILENAME statement, macro variables,
the SET INDSNAME option, and the CALL EXECUTE command within one SAS® Enterprise Guide® Program
node. You have the exciting opportunity to learn how 1,469 time-consuming manual steps are amazingly
replaced with only one time-saving dynamic automated step. We invite you to attend our session, in which we
detail the UNIX and Microsoft Windows syntax for our project example and introduce you to your newest BFF
(Best Friend Forever) in SAS. Please see the appendixes to review additional starting-point information about
the syntax for Windows and IBM z/OS, and to review the source code that created the data sets for our
project example.

INTRODUCTION

SAS is highly regarded around the world, and rightly so, as a powerful, intuitive, and flexible programming
language. As we like to say, SAS enables you to creatively program Smarter And Smarter. However, SAS, as
remarkable as it is, will remain an island unto itself without your increasing coding proficiency.
The tagline for SAS is The Power To Know® and your ‘power to know’ greatly expands with your determination
to communicate more efficiently with the server, PC, or mainframe (referred to as server going forward). The
Power To Know enables The Power To Transform which leads to The Power To Execute ~ but these
powers will quickly go down the drain if you do not continuously learn how to request data more efficiently
and how to automate your SAS world.

Here are 3 questions to ask yourself when designing your SAS program:
 How do I request data more efficiently from the server while protecting the integrity of the data?
 How do I automate my program to eliminate time-consuming and error prone manual processing to gain
back valuable time for more enjoyable SAS endeavors?
 How do I pursue and accomplish this grand and noble deed?

Good News ~ we are going to design a SAS Enterprise Guide program node to:
 Transform a static FILENAME statement into a dynamic FILENAME statement to obtain a Directory Listing
of a date range of files from the server.
 Utilize the Directory Listing to transform Extract, Append, and Export static code into dynamic code.


Dynamic code is executable code based upon parameters that can change, and therefore may or may
not run exactly the same way.



The dynamic code in this presentation re-creates static code which is executable code that never
changes and always runs exactly the same way.



The dynamic code will store the re-created static code in a variable in a SAS dataset.

 Execute the re-created static code automatically with no manual processing or intervention.

The SAS project in this presentation demonstrates:
The Power To Know through a dynamic FILENAME statement and macro variables
The Power To Transform static code into dynamic code using SET INDSNAME option and macro variables
The Power To Execute the re-created static code automatically using the CALL EXECUTE command
We invite you to journey with us as we share how
Dynamic Code
can become your Best Friend Forever in SAS.

 A Tale of SAS Wis-h-dom 
As stated before, the SAS programming language is powerful, intuitive, and flexible. When we wish for a better
way to design our programs, we can tap into the built-in wisdom of SAS. Thus, we have coined the phrase SAS
Wis-h-dom to describe the blending of a SAS wish with SAS wisdom. Discovering the power of combining
dynamic code with a dynamic FILENAME Statement, the SET INDSNAME option, and the CALL EXECUTE
command was as Bob Ross, the well-known painter on PBS so often said, “A happy accident.”
When Bob needed to change his plan for a painting, he referred to the detour as a Happy Accident. Likewise,
when we started the following project with one plan in mind, we soon found that in order to overcome obstacle
bumps on the project road, we needed to discover creative new ways to accomplish the Project Requirements.
On a recent SAS quest, we made several discoveries which we are eager to share with you through our project
example. Read on to learn about the Project Requirements, the SAS Wis-h-dom that transpired along the way,
and the Happy Accidents which occurred on the journey. This project was prompted by a business need to
make the research and analysis of vital variables from 13 years of weekly snapshot data sets more efficient.

Project Requirements:
 Extract vital variables from 52 weekly snapshot data sets per year for 13 years (2006-2018) and combine
them with a Load_Date variable created from the Friday date value derived from the filenames of the data
sets.

 Append the 52 weekly snapshot data sets per year to create 13 yearly data sets.
 Export the 13 appended yearly data sets back to the folder on the server where the weekly snapshot data
sets are stored.
Since SAS Enterprise Guide was being used to design this project, the first decision to make was, “Should the
program be designed with Graphical User Interface (GUI) and/or a program node?

Here are the questions considered in the programming decision, “To GUI or not to GUI?”:
 What will it take to manually add 52 weekly snapshot data sets to the project?
 What will it take to manually create 52 queries to select vital variables from 52 data sets?
 What will it take to manually enter the derived value of the Load_Date variable in 52 queries?
 What will it take to manually append the 52 new data sets created by the 52 queries?
 What will it take to manually export the appended yearly data set back to the server?
 Once the program is designed, what will it take to manually swap 52 inputs and manually update the
Load_Date variable in 52 queries ~ 12 more times ~ while running the program for the 13 year timeframe?

Are

you

getting

tired

yet?

It was determined that the 209 manual steps needed to design the program, and the 105 manual steps
needed to update the program each year, could be done with GUI. However, it also became apparent that the
1,469 manual steps required to run the program for the 13 year timeframe would be excessive and prone to
errors. As a result, our SAS intuition said, “There must be a smarter, easier, and faster way to do this in SAS!”
Dear SAS Wisdom,
We wish we could find a way to automate this program
and eliminate manual processing and intervention ~
except of course for choosing the year.
We look forward to hearing from you soon,
Thank You

By the way, are you in tune with your SAS intuition? Be sure to listen closely when the quiet, reassuring voice
within you says with conviction, “There must be a better way to do this in SAS!” We encourage you to honor
your SAS intuition and to let it motivate you to find new ways to maximize your programming.
“And now for the rest of the story…”,
as Paul Harvey so often said on the radio.

The SAS Quest
Starting
is the first step
towards success.
John C. Maxwell

Sometimes at the beginning of a project it can be challenging to figure out how to accomplish the requirements.
Always remember, the only thing we really need to do is take the first step ~ and the rest will soon follow.

Our first step was to revise the previous programming questions:
 What will it take to automatically create one DATA step to read and append 52 data sets together?
 What will it take to automatically extract vital variables in one DATA step?
 What will it take to automatically enter the derived value of the Load_Date variable in one DATA step?
 What will it take to automatically export the appended yearly data set back to the server?
 Once the program is designed, what will it take to automatically swap 52 inputs and automatically update
the Load_Date variable in one DATA step ~ 12 more times ~ while running the program for the 13 year
timeframe?

 Team Phelps Law 
Everything is easier than it looks;
it will be more rewarding than you expect;
and if anything can go right
~ it will ~
and at the best possible moment.
We began a quest to accomplish the grand and noble deed  of automating this program. Our first task was to
find a way to transform a static FILENAME statement into a dynamic FILENAME statement to read 52 weekly
snapshot data sets from a folder on the server automatically and sequentially ~ rather than manually one at a
time. A Google search led to an article titled Using FILEVAR= To Read Multiple External Files in a DATA Step.

Here is a brief overview of the article:
 The article explained many different ways to transform a static FILENAME statement into a dynamic
FILENAME statement to automatically and sequentially read the content of multiple data sets.
Obstacle Bump ~ Unfortunately this article did not mention how to use a dynamic FILENAME statement to
obtain a Directory Listing of the filename of each data set while reading multiple data sets ~ Bummer!

 Happy Accident Alert  ~ We did not give up and began a series of researching detours. Along the way we
finally discovered that when a dynamic FILENAME statement is used, SAS will actually assign a variable called
FILENAME to the name of each file being read ~ Yea!
This knowledge enabled us to transform a static FILENAME statement into a dynamic FILENAME statement to
obtain a Directory Listing of the filenames that can be utilized to read the content of the files while also
deriving the value of a variable from the filenames of the files being read.
Obstacle Bump ~ However, we then realized even though a dynamic FILENAME statement can be used to
obtain a Directory Listing that can be utilized to transform static code into dynamic code that automatically recreates the static code, we determined the same dynamic FILENAME statement could not be used again within
the re-created static code to obtain the name of each data set as it is actually being read.
Our SAS intuition pondered once again, “Surely, when multiple input data sets are used as inputs in a DATA
step, there must be a way to obtain the name of the data set from where each input observation is read!”

 Happy Accident Alert  ~ Another Google search happily uncovered a SET option called INDSNAME
which identifies the input data set being read with each input observation.
We concluded that a variable called FILENAME can be used to identify the name of an input data set when
using a dynamic FILENAME statement, and a variable called INDSNAME can be used to identify the name of
the input data set when using a SET statement using the SET INDSNAME option.

Learning this information enabled us to design a program to utilize:
 A dynamic FILENAME statement to obtain one Directory Listing per year for 13 years of the filenames of
the 52 weekly snapshot data sets.
 The Directory Listing to transform Extract, Append, and Export static code into dynamic code that
automatically re-creates the static code to:


Extract vital variables from the data sets and combine them with a Load_Date variable created from the
Friday date value derived from the filenames of the data sets using the SET INDSNAME Option.



Append the 52 weekly snapshot data sets per year to create 13 yearly data sets.



Export the 13 appended yearly data sets back to the folder on the server where the weekly snapshot
data sets are stored.

Once the program has run, the re-created Extract, Append, and Export static code can be run manually by
copying and pasting the code into another program node. This program fulfills most of the project
requirements… but remember, our SAS wish was to COMPLETELY automate this project.
As we
continue on
our journey,
we will shed
more light
on this
exciting
SAS
quest.

SAS Illumination
Sometimes success is seeing
what we already have
in a
new light.
Dan Miller

Uh, oh... another Obstacle Bump ~ After we determined how to transform a static FILENAME statement into
a dynamic FILENAME statement to obtain a Directory Listing to utilize in transforming Extract, Append, and
Export static code into dynamic code that automatically re-creates the static code ~ a very important question
arose, “Is there also a way to execute the re-created static code automatically?” You guessed it… our SAS
intuition spoke up again, “There must be a way to call and execute a variable in a SAS data set containing a SAS
DATA step.”
 Happy Accident Alert  ~ It only took one more hopeful Google search to find a White Paper titled CALL
EXECUTE: A Powerful Data Management Tool which revealed that a CALL EXECUTE command already existed!

Here is a brief overview of the White Paper:
 CALL EXECUTE (variable); resolves and executes the value of a variable.
 The variable can be a character variable in a data set containing SAS statements such as a DATA step.
 The CALL EXECUTE command will execute the re-created static code automatically and will enable us to
finally fulfill all of the project requirements!

Yea!!!
Strike up the band,
Toss the confetti,
Release the balloons!
Applause…

Applause…

Applause…

Bring out the treats for everyone!

SAS illumination ~ we will use a dynamic FILENAME statement to obtain a Directory Listing to utilize in
transforming Extract, Append, and Export static code into dynamic code that automatically re-creates the static
code, and then use the CALL EXECUTE command to execute the static code automatically. The program will
run automatically without any manual processing or intervention ~ except for choosing the year!

Here is the program displayed as a SAS Enterprise Guide program node:

Create

Create
Yearly
Yearly
data sets
sets
data

As you can see from this SAS quest, it pays to listen to your SAS intuition. Our determination to overcome
obstacles ~ and a series of simple Google searches ~ led to resources which illuminated how to fulfill the
project requirements and enabled this project to become a very successful reality. Always remember there is a
treasure chest of SAS information waiting on the web to help you maximize the quality, efficiency, and
automation of your programming.
On the next leg of our journey
we will walk you through a
step-by-step demonstration of

The Power To Know, Transform, and Execute

The first step is the most important step you will take ~
and the last step is the most rewarding step you will experience.
Kent  Ronda Team Phelps

THE POWER TO KNOW
Through a Dynamic FILENAME Statement
Disclaimer: This presentation details the UNIX and Microsoft Windows syntax for our project example.
Refer to your specific Operating System (e.g. UNIX, Windows, or IBM z/OS) Manual, Installation
Configuration, and/or in-house Technical Support for further guidance in how to create the SAS code
presented in this paper. Please see Appendix A for additional starting-point information regarding
the syntax for Windows and z/OS.
The following example highlights how to transform a static FILENAME into a dynamic FILENAME statement
to obtain a Directory Listing of the 52 weekly data sets for the year 2018 from a folder on the server.

This code will obtain and store the Directory Listing:
Create
Yearly
data sets

FILENAME indata '/SMILEY/file2018*.sas7bdat';
DATA path_list_files;
LENGTH fpath SAS_data_set_and_path $100;
RETAIN fpath;
INFILE indata TRUNCOVER FILENAME=SAS_data_set_and_path;
INPUT;
IF fpath NE SAS_data_set_and_path THEN;
DO;
fpath = SAS_data_set_and_path;
...
OUTPUT;
END;
RUN;

 This code will obtain a Directory Listing of the data sets following the file2018*.sas7bdat pattern
from the SMILEY folder on the server and store it in a data set.

These are the 3 weekly data sets being processed in our example:
file20180105.sas7bdat
file20180112.sas7bdat
file20180119.sas7bdat

 Each must follow the same pattern fileYYYYMMDD.sas7bdat (See Appendix B for code to create).

This is a sample of the columns and formatting for each data set:

 This contains each Special Person, Number, and Code for employees of the  Smiley Company .

Creating a dynamic FILENAME statement:
FILENAME indata '/SMILEY/file2018*.sas7bdat';

 The FILENAME statement assigns indata as a file reference (fileref) to the folder and file pattern.
 The asterisk within the file pattern file2018*.sas7bdat transforms the static FILENAME statement
into a dynamic FILENAME statement which will read multiple files automatically and sequentially.
 The FILENAME=variable statement assigns the path and name of each file being read.
 In summary, a dynamic FILENAME statement and the FILENAME=variable statement will obtain the
Directory Listing.

Creating a DATA step to read and store the Directory Listing:
DATA path_list_files;
LENGTH fpath SAS_data_set_and_path $100;
RETAIN fpath;

 The DATA statement creates an output data set called path_list_files.
 The LENGTH statement assigns a length of 100 characters to a variable that will store each unique
data set path and filename called fpath.
 The RETAIN statement retains the value of fpath until it is assigned a new filename later in the code.
 The LENGTH statement also assigns a length of 100 characters to a variable that will be used to store
and track changes to the data set path and filename called SAS_data_set_and_path.
 In summary, the path_list_files data set is created to contain the 100 character fpath and
SAS_data_set_and_path variables which will be used to read and store the Directory Listing.

Preparing the INFILE indata (fileref) for use and the INPUT of data:
INFILE indata TRUNCOVER FILENAME=SAS_data_set_and_path;
INPUT;

 The INFILE statement assigns indata to be read with the INPUT statement.
 The TRUNCOVER option tells SAS the input data may or may not be the same length.
 The FILENAME=SAS_data_set_and_path statement assigns SAS_data_set_and_path to the path
and filename being read.
 The INPUT statement reads the INFILE indata (fileref) sequentially without creating any variables.
 In summary, INFILE assigns indata to be read with an INPUT of variable length (without creating
any variables) while assigning SAS_data_set_and_path to each path and filename being read.

Creating an IF-THEN DO-END statement to detect new filenames being read:
IF fpath NE SAS_data_set_and_path THEN;
DO;
fpath = SAS_data_set_and_path;
...
OUTPUT;
END;

 The IF-THEN statement executes the contents of the DO-END when a new filename is read.
 The fpath = SAS_data_set_and_path statement assigns the fpath variable to the value of the
SAS_data_set_and_path variable which contains the path and filename as each new file is read.
 The OUTPUT statement is executed within the IF-THEN DO-END statement to ensure that we only
write an observation re-creating static code when a new file is read and fpath changes.
 In summary, the fpath variable is assigned to the path and filename of each new data set (Directory
Listing) up to 100 characters as the filename of the data sets change.

Here are the statements combined with a RUN statement:
FILENAME indata '/SMILEY/file2018*.sas7bdat';
DATA path_list_files;
LENGTH fpath SAS_data_set_and_path $100;
RETAIN fpath;
INFILE indata TRUNCOVER FILENAME=SAS_data_set_and_path;
INPUT;
IF fpath NE SAS_data_set_and_path THEN;
DO;
fpath = SAS_data_set_and_path;
...
OUTPUT;
END;
RUN;

This is the output data set created by the preceding statements:
/SMILEY/file20180105.sas7bdat
/SMILEY/file20180112.sas7bdat
/SMILEY/file20180119.sas7bdat

 Next we will explore how this Directory Listing is used to transform static code into dynamic code.

9

THE POWER TO TRANSFORM
Static Code into Dynamic Code Using the SET INDSNAME Option
The following example highlights how to transform Extract, Append, and Export static code into dynamic code
that automatically recreates the static code to Extract vital variables from 52 weekly data sets and combine
with a Load_Date variable (derived from the filenames of the data sets), how to Append the 52 weekly data
sets to create a yearly data set, and how to Export the yearly data set to the server.

This is the original Extract static code for weeks 1 and 3:
DATA file_final_20180105;
SET '/SMILEY/file20180105.sas7bdat';
FORMAT Load_Date date9.; Load_Date = '05JAN2018'd;
KEEP Special_Person Special_Number Load_Date;
RUN;
DATA file_final_20180119;
SET '/SMILEY/file20180119.sas7bdat';
FORMAT Load_Date date9.; Load_Date = '19JAN2018'd;
KEEP Special_Person Special_Number Load_Date;
RUN;

 Each weekly DATA step creates a file_final_YYYYMMDD data set with the Load_Date variable
derived and formatted as a SAS date (date9) from the YYYYMMDD create date of the data set.
 A KEEP statement keeps the Special_Person, Special_Number and the derived Load_Date.

Here is the Append code combined with the Export code for weeks 1 to 3:
DATA '/SMILEY/file_final_2018.sas7bdat';
SET file_final_20180105.sas7bdat
file_final_20180112.sas7bdat
file_final_20180119.sas7bdat;
RUN;

 Each Append DATA step creates a file_final_YYYY data set with YYYY matching the year.
 Each of the file_final_YYYYMMDD data sets are SET as data sets one after another.

Here is the Extract, Append, and Export code almost completely combined:
DATA '/SMILEY/file_final_2018.sas7bdat';
SET '/SMILEY/file20180105.sas7bdat'
'/SMILEY/file20180112.sas7bdat'
'/SMILEY/file20180119.sas7bdat';
FORMAT Load_Date date9.; Load_Date = '05JAN2018'd;
KEEP Special_Person Special_Number Load_Date;
RUN;

 The major change from the previous Append is that this Append SETs the 52 original data sets as
inputs rather than using the outputs from the 52 Extract DATA steps.
 Question: Since the dynamic FILENAME statement, used to execute the dynamic code that recreates
this static code, is not available during the runtime of this static code, how do we obtain Load_Date?

Here is the Extract, Append, and Export code completely combined:
DATA '/SMILEY/file_final_2018.sas7bdat';
LENGTH Current_File $100;
SET '/SMILEY/file20180105.sas7bdat'
'/SMILEY/file20180112.sas7bdat'
'/SMILEY/file20180119.sas7bdat'
INDSNAME=Current_File;
FORMAT Load_Date date9.;
KEEP Special_Person Special_Number Load_Date;
Load_Date = MDY(INPUT(SUBSTR(Current_File,17,2),2.),
INPUT(SUBSTR(Current_File,19,2),2.),
INPUT(SUBSTR(Current_File,13,4),4.));
RUN;

 Create a variable Current_File with a LENGTH long enough for each data set name and path in the
Directory Listing.
 Place the LENGTH before the SET statement so the complete data set name and path are captured.
 Add INDSNAME=Current_File to the end of the SET statement so Current_File will always be
assigned the data set name and path of the observation being read.
 Use the MDY, INPUT, and SUBSTR functions to transform the month, day, and year of each data set
name and path (Current_File) into the Load_Date.

Here is the Extract, Append, and Export code efficiently combined:
DATA '/SMILEY/file_final_2018.sas7bdat';
LENGTH Current_File Prior_File $100;
SET '/SMILEY/file20180105.sas7bdat'
'/SMILEY/file20180112.sas7bdat'
'/SMILEY/file20180119.sas7bdat'
INDSNAME=Current_File;
FORMAT Load_Date date9.;
KEEP Special_Person Special_Number Load_Date;
RETAIN Load_Date;
IF Current_File NE Prior_File THEN
DO;
Load_Date = MDY(INPUT(SUBSTR(Current_File,17,2),2.),
INPUT(SUBSTR(Current_File,19,2),2.),
INPUT(SUBSTR(Current_File,13,4),4.));
Prior_File = Current_File;
END;
RUN;

 The previous DATA step correctly derives and assigns the Load_Date with each observation.
 However, for better efficiency, we create another variable Prior_File and an IF Current_File
NE Prior_File THEN statement to track data set changes and assign Load_Date with each change.
 Then we add a RETAIN Load_Date statement to retain the Load_Date when the data set does not
change.
 The next step is to transform the efficient static code into dynamic code.

Begin to transform static code into dynamic code using quotes and timing:
FILENAME indata '/SMILEY/file2018*.sas7bdat';
DATA path_list_files;
LENGTH SAS_data_set_and_path fpath $100 fpath_line $10000;
RETAIN fpath;
FORMAT Load_Date date9.;
INFILE indata TRUNCOVER FILENAME=SAS_data_set_and_path;
INPUT;
IF fpath NE SAS_data_set_and_path THEN
DO;
fpath = SAS_data_set_and_path;
fpath_line = CATS("
DATA '/SMILEY/file_final_","2018",".sas7bdat';
LENGTH Current_File Prior_File $100;
SET '","/SMILEY/file20180105.sas7bdat","'
'","/SMILEY/file20180112.sas7bdat","'
'","/SMILEY/file20180119.sas7bdat","'
INDSNAME=Current_File;
FORMAT Load_Date date9.;
KEEP Special_Person Special_Number Load_Date;
RETAIN Load_Date;
IF Current_File NE Prior_File THEN
DO;
Load_Date = MDY(INPUT(SUBSTR(Current_File,17,2),2.),
INPUT(SUBSTR(Current_File,19,2),2.),
INPUT(SUBSTR(Current_File,13,4),4.));
Prior_File = Current_File;
END;
RUN;");
OUTPUT;
END;
RUN;

Use quotation marks to surround the static code and what changes within:
 Create a variable fpath_line long enough to contain the concatenation with spaces removed (CATS)
of the static code in quotation marks.
 Surround the static code with quotation marks to begin the process of transforming the code.
 If single quotes are contained within the static code, use double quotes to surround the static code.
 Surround the year of the output data set with double quotes and commas because it will be derived
from the year of the input data sets.
 Surround the names of the input data sets with double quotes and commas because they will be
derived from the names of the input data sets.

Identify the timing of changes with each observation of static code:
 The output data set year 2018 within the DATA through SET statements will be derived from the first
observation of fpath that is derived from SAS_data_set_and_path.
 The /SMILEY/file20180105.sas7bdat through /SMILEY/file20180119.sas7bdat input data set
file names will be derived from each observation of fpath.
 The INDSNAME=Current_File; through RUN; statements will occur at the end after all observations
of the Directory Listing have been read.

Code for the timing of changes with each observation of static code:
FILENAME indata '/SMILEY/file2018*.sas7bdat';
DATA path_list_files;
LENGTH SAS_data_set_and_path fpath $100 fpath_line $10000;
RETAIN fpath fpath_line;
FORMAT Load_Date date9.;
INFILE indata TRUNCOVER FILENAME=SAS_data_set_and_path END=DONE;
INPUT;
IF _N_ = 1 THEN
fpath_line =
CATS("DATA '/SMILEY/file_final_",
SUBSTR(SAS_data_set_and_path,13,4),".sas7bdat';
LENGTH Current_File Prior_File $100;
SET ");
IF fpath NE SAS_data_set_and_path THEN
DO;
fpath = SAS_data_set_and_path;
fpath_line = CATS(fpath_line)||" '"||CATS(fpath)||"' ";
END;
IF DONE THEN
DO;
fpath_line = CATS(fpath_line)||
"INDSNAME=Current_File;
FORMAT Load_Date date9.;
KEEP Special_Person Special_Number Load_Date;
RETAIN Load_Date;
IF Current_File NE Prior_File THEN
DO;
Load_Date = MDY(INPUT(SUBSTR(Current_File,17,2),2.),
INPUT(SUBSTR(Current_File,19,2),2.),
INPUT(SUBSTR(Current_File,13,4),4.));
Prior_File = Current_File;
END;
RUN;";
OUTPUT;
END;
RUN;

 Add the fpath_line variable to the RETAIN statement to enable each piece of the DATA step to be
concatenated to the fpath_line variable until the result is the complete dynamic code.
 Add IF _N_ = 1 THEN to begin what must occur with the first observation and change 2018 to the
derivation of SUBSTR(SAS_data_set_and_path,13,4) noting that SAS_data_set_and_path is
assigned the name of each input data set and path by the FILENAME=SAS_data_set_and_path
option on the INFILE statement.
 Add the IF fpath NE SAS_data_set_and_path THEN logic so that the assignment of fpath =
SAS_data_set_and_path only occurs when the first observation of each new input data set is read.
 Assign fpath_line to itself and remove the beginning and ending spaces ( CATS(fpath_line)) and
concatenate (||) it to each input data set and path (CATS(fpath)) surrounded by single quotes (').
 Add END=DONE to the INFILE statement so DONE will be set to True and the IF DONE THEN will
execute after the last observation is read from the last input data set.
 Assign fpath_line to itself and remove the beginning and ending spaces ( CATS(fpath_line)) and
concatenate (||) it to the remainder of the code needed to complete the DATA step.
 Add the OUTPUT statement to create only one observation of the fpath_line variable containing the
complete and executable dynamic code.

This is the dynamic code which recreates the original static code:
FILENAME indata '/SMILEY/file2018*.sas7bdat';
DATA path_list_files;
LENGTH SAS_data_set_and_path fpath $100 fpath_line $10000;
RETAIN fpath fpath_line;
FORMAT Load_Date date9.;
INFILE indata TRUNCOVER FILENAME=SAS_data_set_and_path END=DONE;
INPUT;
IF _N_ = 1 THEN
fpath_line = CATS("DATA '/SMILEY/file_final_",
SUBSTR(SAS_data_set_and_path,13,4),".sas7bdat';
LENGTH Current_File Prior_File $100;
SET ");
IF fpath NE SAS_data_set_and_path THEN
DO;
fpath = SAS_data_set_and_path;
fpath_line = CATS(fpath_line)||" '"||CATS(fpath)||"' ";
END;
IF DONE THEN
DO;
fpath_line = CATS(fpath_line)||
"INDSNAME=Current_File;
FORMAT Load_Date date9.;
KEEP Special_Person Special_Number Load_Date;
RETAIN Load_Date;
IF Current_File NE Prior_File THEN
DO;
Load_Date = MDY(INPUT(SUBSTR(Current_File,17,2),2.),
INPUT(SUBSTR(Current_File,19,2),2.),
INPUT(SUBSTR(Current_File,13,4),4.));
Prior_File = Current_File;
END;
RUN;";
OUTPUT;
END;
RUN;

 We now add two macro variables to enhance the processing of the input and output file year and the
viewing of the code deriving the Load_Date from the Month, Day, and Year of the Current_File.

Assign a macro variable to the file year:
%LET File_Year = 2018;
FILENAME indata "/SMILEY/file&File_Year*.sas7bdat";
...
IF _N_ = 1 THEN
fpath_line = 'DATA "/SMILEY/file_final_&File_Year..sas7bdat";
LENGTH Current_File Prior_File $100;
SET ';
...
RUN;

 Add %LET to assign a new macro variable called File_Year equal to 2018.
 Replace 2018 in the FILENAME statement with the macro assignment &File_Year and change the
single quotes (') to double quotes (") around the input file name to resolve the macro variable.
 Replace SUBSTR(SAS_data_set_and_path,13,4) with the macro &File_Year. ~ include a period
after &File_Year to delimit macro and keep period prior to sas7bdat ~ and change single quotes
(') to double quotes (") around the output file name to resolve the macro variable.
 Change the double quotes ( ") to single quotes (') around the DATA and SET statements in order to
fully contain what is in double quotes (") within the single quotes (').

Assign a macro variable to the current file month, day, and year:
%LET Current_File_MDY = INPUT(SUBSTR(Current_File,17,2),2.),
INPUT(SUBSTR(Current_File,19,2),2.),
INPUT(SUBSTR(Current_File,13,4),4.);
...
IF DONE THEN
DO;
fpath_line = CATS(fpath_line)||
"INDSNAME=Current_File;
...
IF Current_File NE Prior_File THEN
DO;
Load_Date = MDY(&Current_File_MDY);
Prior_File = Current_File;
END;
RUN;";
...
RUN;

 Add the %LET to assign a new macro variable called Current_File_MDY to the three lines of code in
the parenthesis of the MDY function of the Load_Date assignment.
 Replace the code in the parenthesis of the MDY function with the macro &Current_File_MDY.
 Since double quotes (") already surround this section of code, no further changes are necessary.

This is the final dynamic code which recreates the original static code:
%LET File_Year = 2018;
%LET Current_File_MDY = INPUT(SUBSTR(Current_File,17,2),2.),
INPUT(SUBSTR(Current_File,19,2),2.),
INPUT(SUBSTR(Current_File,13,4),4.);
FILENAME indata "/SMILEY/file&File_Year*.sas7bdat";
DATA path_list_files;
LENGTH SAS_data_set_and_path fpath $100 fpath_line $10000;
RETAIN fpath fpath_line;
FORMAT Load_Date date9.;
INFILE indata TRUNCOVER FILENAME=SAS_data_set_and_path END=DONE;
INPUT;
IF _N_ = 1 THEN
fpath_line = 'DATA "/SMILEY/file_final_&File_Year..sas7bdat";
LENGTH Current_File Prior_File $100;
SET ';
IF fpath NE SAS_data_set_and_path THEN
DO;
fpath = SAS_data_set_and_path;
fpath_line = CATS(fpath_line)||" '"||CATS(fpath)||"' ";
END;
IF DONE THEN
DO;
fpath_line = CATS(fpath_line)||
"INDSNAME=Current_File;
FORMAT Load_Date date9.;
KEEP Special_Person Special_Number Load_Date;
RETAIN Load_Date;
IF Current_File NE Prior_File THEN
DO;
Load_Date = MDY(&Current_File_MDY);
Prior_File = Current_File;
END;
RUN;";
OUTPUT;
END;
RUN;

THE POWER TO EXECUTE
Dynamic Code Automatically Using the CALL EXECUTE Command
After transforming the static code into dynamic code that automatically recreates the static code to Extract,
Append, and Export the yearly data set, the CALL EXECUTE command will execute the code automatically.

CALL EXECUTE the dynamic code to recreate and run the static code:
DATA _NULL_;
SET path_list_files;
CALL EXECUTE(fpath_line);
RUN;

 The DATA step does not create an output data set because the _NULL_ option is used.
 The SET statement sets path_list_files as the input data set for this DATA step.
 The CALL EXECUTE command runs the fpath_line variable in the path_list_files data set.

Here is the code with the fpath_line variable resolved as dynamic code:
DATA _NULL_;
SET path_list_files;
CALL EXECUTE(
%LET File_Year = 2018;
%LET Current_File_MDY = INPUT(SUBSTR(Current_File,17,2),2.),
INPUT(SUBSTR(Current_File,19,2),2.),
INPUT(SUBSTR(Current_File,13,4),4.);
FILENAME indata "/SMILEY/file&File_Year*.sas7bdat";
DATA path_list_files;
LENGTH SAS_data_set_and_path fpath $100 fpath_line $10000;
RETAIN fpath fpath_line;
FORMAT Load_Date date9.;
INFILE indata TRUNCOVER FILENAME=SAS_data_set_and_path END=DONE;
INPUT;
IF _N_ = 1 THEN fpath_line = 'DATA "/SMILEY/file_final_&File_Year..sas7bdat";
LENGTH Current_File Prior_File $100;
SET ';
IF fpath NE SAS_data_set_and_path THEN
DO; fpath = SAS_data_set_and_path;
fpath_line = CATS(fpath_line)||" '"||CATS(fpath)||"' ";
END;
IF DONE THEN
DO; fpath_line = CATS(fpath_line)||
"INDSNAME=Current_File;
FORMAT Load_Date date9.;
KEEP Special_Person Special_Number Load_Date;
RETAIN Load_Date;
IF Current_File NE Prior_File THEN
DO;
Load_Date = MDY(&Current_File_MDY);
Prior_File = Current_File;
END;
RUN;";
OUTPUT;
END;
RUN;

);
RUN;

Here is the code with the dynamic code resolved as the original static code:
DATA _NULL_;
SET path_list_files;
CALL EXECUTE(
DATA '/SMILEY/file_final_2018.sas7bdat';
LENGTH Current_File Prior_File $100;
SET '/SMILEY/file20180105.sas7bdat'
'/SMILEY/file20180112.sas7bdat'
'/SMILEY/file20180119.sas7bdat'
INDSNAME=Current_File;
FORMAT Load_Date date9.;
KEEP Special_Person Special_Number Load_Date;
RETAIN Load_Date;
IF Current_File NE Prior_File THEN
DO;
Load_Date = MDY(INPUT(SUBSTR(Current_File,17,2),2.),
INPUT(SUBSTR(Current_File,19,2),2.),
INPUT(SUBSTR(Current_File,13,4),4.));
Prior_File = Current_File;
END;
RUN;
);
RUN;

 Here is the result of fully executing fpath_line in the path_list_files data set:
FILE_ALL_2018

05JAN2018
05JAN2018
05JAN2018
05JAN2018
05JAN2018
12JAN2018
12JAN2018
12JAN2018
12JAN2018
12JAN2018
19JAN2018
19JAN2018
19JAN2018
19JAN2018
19JAN2018

Now that we have completed the process for one year, we need to repeat the process for the remaining 12
years of this project. How is this accomplished? We simply update the year in the File_Year macro
variable in the Create Yearly data set program node, rerun the program node, and then repeat this process
until each remaining year is complete.

Creating the yearly data sets for each year:
%LET File_Year = 2018;

 Update the File_Year macro variable in the program node and rerun for each year.

 Done and Done 

CONCLUSION
The Power To Know through a dynamic FILENAME statement and macro variables enables The Power To
Transform static code into dynamic code using the SET INDSNAME option and macro variables which leads
to The Power To Execute the re-created static code automatically using the CALL EXECUTE command.
{Try saying that statement really fast for fun!} You have seen how 1,469 time-consuming manual steps
are amazingly replaced with only one time-saving dynamic automated step.
On your future SAS quests, listen closely to your SAS intuition and pursue blending your SAS wishes with the
built-in wisdom of SAS. As you experience SAS Wis-h-dom, your research will lead you to your own Happy
Accident discoveries which will increase the efficiency and automation of your program designs. As you leave
here with your newest BFF in SAS, begin thinking about how you can benefit from this powerful partnership.

It’s not what the world holds for you,
it’s what YOU bring to it!
Anne of Green Gables

It’s not what the SAS world holds for you, it’s what YOU bring to it. You are like the language itself ~ you are
intuitive and flexible in designing your programs. As a SAS professional, you are inquisitive, research
oriented, and solution driven. Your optimistic and tenacious desire to design a quality program fuels your
thoroughness and attention to detail. When you are in your SAS zone, you are relentless in your pursuit to
overcome obstacles and maximize your programming.
Don’t be a reservoir, be a river. John C. Maxwell
SAS programming is Mind Art ~ a creative realm where each of you is an artist. Continue to develop and build
on your many skills and talents. Keep looking for different ways to share your God-given abilities and ideas.
Don’t be a reservoir of SAS knowledge, be a river flowing outward to empower those around you!
Your life is like a campfire at night ~
You never know how many people will see it
and be comforted and guided by your light.
Claire Draper

Always remember, your contributions make a positive impact in the world. Plan on coming back to SAS
Global Forum next year to shed some light on the exciting things you are learning. All of us are on the SAS
journey with you and we look forward to your teaching sessions in the future.
As we conclude, we want to introduce you to our SAS mascot, Smiley. Smiley represents the SAS joy which
each of us experience as we find better ways to accomplish mighty and worthy deeds using SAS. We hope we
have enriched your SAS knowledge. You may not use this powerful partnership on a daily basis, but when the
need arises ~ Oh, how powerful and valuable your relationship will be with your newest BFF in SAS!
Thank You for sharing part of your SAS journey with us ~
 Happy SAS trails to you… until we meet again 

18

ACKNOWLEDGMENTS
We want to thank the diligent White Paper Reviewers of SAS Global Forum 2018 for accepting our abstract
and paper, and we want to express our appreciation to our Interim Presenter Coordinator, Kerri Roselli, and
our Presenter Coordinator, John Harden, for their kind and dedicated assistance to us and all of the attendees,
and to the SAS Global Forum 2018 Chair, Dr. Goutam Chakraborty, Executive Board, Conference Leaders, and
SAS Institute for their comprehensive efforts in organizing this exciting and illuminating conference. You
inspire us to share what we are learning and we hope to be a light of encouragement to you as well ~ Your
friends, Kent  Ronda Team Phelps ~ The SASketeers ~ All for SAS & SAS for All!~ Illuminator Coaching, Inc.

MEET THE AUTHORS
Writing is a permanent legacy.
John C. Maxwell

Kent & Ronda TEAM Phelps are the Co-Founders of Illuminator Coaching, Inc., & The SASketeers: All for SAS
and SAS for All! ~ they have co-authored 12 SAS white papers, studied Transformational Leadership, Dynamic
Teamwork, and Personal Growth for over 20 years, and are certified as John Maxwell Team Coaches and 48
Days To The Work You Love Coaches.
Kent wants to encourage and equip you to fulfill your life, career, and leadership potential as you build an
enduring legacy of inspiration, excellence, and honor ~ SAS® Certified Professional Programmer Consultant ~
B.S. Electrical Engineering ~ serving for over 20 years as an essential bridge builder of consensus and quality
programming to connect the needs of Business and IT ~ happily programmed in Base SAS® and SAS®
Enterprise Guide® since 2007 with a strong focus on engineering innovative, efficient, and automated
solutions ~ presented at the MidWest SAS® Users Group (MWSUG) Conference for the last five years including
two Hands-On Workshops.
Ronda believes that YOU are a gift the world is waiting to receive, and she wants to encourage and equip you
to pursue your unique destiny as you navigate your life journey with intentionality, fulfilling purpose, and
enduring hope ~ Writer & Coach ~ gifted in helping others to explore and express their hearts and minds
through writing ~ served in the Banking and Insurance industries for 19 years ~ presented at two MidWest
SAS® Users Group (MWSUG) Conferences.
We invite you to share your valued comments with us:
Kent  Ronda Team Phelps
The SASketeers ~ All for SAS & SAS for All!
E-mail: SASketeers@IlluminatorCoaching.com

 We look forward to connecting with you in the future! 

REFERENCES
Agarwal, Megha (2012), The Power of “The FILENAME” Statement, Gilead Sciences, Foster City, CA, USA.
http://www.lexjansen.com/wuss/2012/63.pdf
Gan, Lu (2012), Using SAS® to Locate and Rename External Files, Pharmaceutical Product Development, L.L.C., Austin, TX, USA.
http://www.scsug.org/wp-content/uploads/2012/11/Using-SAS-to-locate-and-rename-external-files.pdf
Hamilton, Jack (2012), Obtaining a List of Files in a Directory Using SAS® Functions.
http://www.wuss.org/proceedings12/55.pdf

19

Lafler, Kirk Paul and Charles Edwin Shipp (2012), Google® Search Tips and Techniques for SAS® and JMP® Users, Proceedings of the 23rd
Annual MidWest SAS Users Group (MWSUG) 2012 Conference, Software Intelligence Corporation, Spring Valley, CA, and Consider
Consulting, Inc., San Pedro, CA, USA.
http://www.mwsug.org/proceedings/2012/JM/MWSUG-2012-JM06.pdf
Langston, Rick (2013), Submitting SAS® Code On The Side; SAS Institute Inc., Cary, NC.
http://support.sas.com/resources/papers/proceedings13/032-2013.pdf
Michel, Denis (2005), CALL EXECUTE: A Powerful Data Management Tool, Proceedings of the 30th Annual SAS® Users Group International
(SUGI) 2005 Conference, Johnson & Johnson Pharmaceutical Research and Development, L.L.C.
http://www2.sas.com/proceedings/sugi30/027-30.pdf
Phelps, Kent  Ronda Team (2017), Hands-On Workshop (HOW): Base SAS® and SAS® Enterprise Guide® ~ Automate Your SAS World
With Dynamic Code; Your Newest BFF (Best Friend Forever) in SAS, Proceedings of the 28th Annual MidWest SAS Users Group (MWSUG)
2017 Conference, The SASketeers ~ All for SAS and SAS for All! ~ Illuminator Coaching, Inc.
https://www.mwsug.org/proceedings/2017/HW/MWSUG-2017-HW03.pdf
Phelps, Kent  Ronda Team (2016), Base SAS® and SAS® Enterprise Guide® ~ Automate Your SAS World With Dynamic Code; Your
Newest BFF (Best Friend Forever) in SAS, Proceedings of the 27th Annual MidWest SAS Users Group (MWSUG) 2016 Conference, The
SASketeers ~ All for SAS and SAS for All! ~ Illuminator Coaching, Inc.
http://www.mwsug.org/proceedings/2016/TT/MWSUG-2016-TT11.pdf
Phelps, Kent  Ronda Team (2016), Hands-On Workshop (HOW): The Joinless Join ~ The Impossible Dream Come True; Expand the
Power of Base SAS® and SAS® Enterprise Guide® in a New Way, Proceedings of the 27th Annual MidWest SAS Users Group (MWSUG) 2016
Conference, The SASketeers ~ All for SAS and SAS for All! ~ Illuminator Coaching, Inc.
http://www.mwsug.org/proceedings/2016/HW/MWSUG-2016-HW03.pdf
SAS Institute Inc. (2017), SAS® 9.4 Companion for z/OS, Sixth Edition; Cary, NC; SAS Institute Inc.
http://documentation.sas.com/?docsetId=hosto390&docsetTarget=titlepage.htm&docsetVersion=9.4&locale=en
SAS Institute Inc. (2017), SAS(R) 9.4 Macro Language: Reference, Fifth Edition; Cary, NC; SAS Institute Inc.
http://documentation.sas.com/?docsetId=mcrolref&docsetTarget=titlepage.htm&docsetVersion=9.4&locale=en
SAS Institute Inc. (2018), SAS(R) 9.4 Statements: Reference, Fourth Edition; Cary, NC; SAS Institute Inc.
http://documentation.sas.com/?docsetId=lestmtsref&docsetTarget=titlepage.htm&docsetVersion=9.4&locale=en
Spector, Phil, An Introduction to the SAS System; Statistical Computing Facility; University of California, Berkeley.
http://www.stat.berkeley.edu/~spector/
Support.SAS.com (2007), Using FILEVAR= to Read Multiple External Files in a DATA Step.
http://support.sas.com/techsup/technote/ts581.pdf
Watson, Richann (2013), Let SAS® Do Your DIRty Work, Experis, Batavia, OH.
http://www.pharmasug.org/proceedings/2013/TF/PharmaSUG-2013-TF06.pdf

TRADEMARK CITATIONS
SAS and all other SAS Institute, Inc., product or service names are registered trademarks or trademarks of SAS
Institute, Inc., in the USA and other countries. The symbol, ®, indicates USA registration. Other brand and
product names are registered trademarks or trademarks of their respective companies.

DISCLAIMER
We have endeavored to provide accurate and helpful information in this SAS Presentation/Workshop/White
Paper. The information is provided in ‘Good Faith’ and ‘As Is’ without any kind of warranty, either expressed
or implied. Recipients acknowledge and agree that we and/or our company are/is not, and never will be,
liable for any problems and/or damages whatsoever which may arise from the recipient’s use of the
information in this paper. Please refer to your specific Operating System (e.g. UNIX, Microsoft Windows, or
IBM z/OS) Manual, Installation Configuration, and/or in-house Technical Support for further guidance in how
to create the SAS code presented in this paper.

Copyright © Kent  Ronda Team Phelps ~ The SASketeers ~ Illuminator Coaching, Inc. ~ All Rights Reserved

20

